

Object subclass: #VRMLObject

  instanceVariableNames: ''

  classVariableNames: ''

  poolDictionaries: ''   



VRMLObject subclass: #VRMLNode

  instanceVariableNames: 

    'name parent fields '

  classVariableNames: ''

  poolDictionaries: '' 



VRMLObject subclass: #VRMLReader

  instanceVariableNames: 

    'isVersion1 variables '

  classVariableNames: ''

  poolDictionaries: '' 



VRMLObject subclass: #VRMLVariable

  instanceVariableNames: 

    'name value isDefinition '

  classVariableNames: ''

  poolDictionaries: ''    







VRMLObject class methods 



fileOutAll

    "VRMLObject fileOutAll"

    Object

        fileOutOn: 'vrml.#2'

        classes: #(VRMLObject)

        methods: #(

            (instance exists: Collection)

            (class fileOutOn:classes:methods:pools: Object))

        pools: #()



new

    ^super new initialize 







VRMLObject methods 

  

initialize

    "Default is to do nothing."



printOn: aStream

    aStream cr.

    self privatePrint: self on: aStream indentNext: 3

   

privateObjectPrintingSelector: object

    (object isKindOf: VRMLObject)

        ifTrue: [^#privatePrintVRMLObject:on:indentNext:].

    object isCollection not | object isString

        ifTrue: [^#privatePrintBasicObject:on:indentNext:].

    (object exists: [:element | (element isKindOf: VRMLObject) not])

        ifTrue: [^#privatePrintBasicObjects:on:indentNext:]

        ifFalse: [^#privatePrintVRMLObjects:on:indentNext:]



privatePrint: object on: aStream indentNext: indent

    self

        perform: (self privateObjectPrintingSelector: object)

        with: object

        with: aStream

        with: indent



privatePrintBasicObject: object on: aStream indentNext: indent

    "This method is private to privatePrint:on:indentNext:."

    object printOn: aStream.



privatePrintBasicObjects: objects on: aStream indentNext: indent

    "This method is private to privatePrint:on:indentNext:."

    aStream nextPutAll: ' ['.

    aStream cr; next: indent put: $ .

    objects do: [:object |

        object printOn: aStream. aStream space].

    aStream cr; next: indent - 3 put: $ ; nextPut: $].



privatePrintVRMLObject: object on: aStream indentNext: indent

    "This method is private to privatePrint:on:indentNext:."

    object printOn: aStream indentNext: indent

   

privatePrintVRMLObjects: objects on: aStream indentNext: indent

    "This method is private to privatePrint:on:indentNext:."

    aStream nextPutAll: ' ['.

    objects do: [:object |

        aStream cr; next: indent put: $ .

        object printOn: aStream indentNext: indent + 3].

    aStream cr; next: indent - 3 put: $ ; nextPut: $]. 







VRMLNode class methods  







VRMLNode methods 

 

at: fieldName

    ^fields at: fieldName

   

at: fieldName ifAbsent: aBlock

    ^fields at: fieldName ifAbsent: aBlock

 

at: fieldName put: anObject

    fields at: fieldName put: anObject.

    (anObject isKindOf: VRMLNode)

        ifTrue: [anObject parent: self]

   

initialize

    name := 'Unknown'.

    fields := IdentityDictionary new

   

name

    ^name



name: aSymbol

    name := aSymbol

 

parent

    ^parent



parent: aVRMLObject

    parent := aVRMLObject

 

printOn: aStream indentNext: indent

    | fieldName fieldValue |

    aStream nextPutAll: self name; space; nextPut: ${.

    fields associationsDo: [:association |

        fieldName := association key. fieldValue := association value.

        aStream cr; next: indent put: $ .

        aStream nextPutAll: fieldName; nextPutAll: ' = '.

        self privatePrint: fieldValue on: aStream indentNext: indent + 3].

    aStream cr; next: indent - 3 put: $ ; nextPut: $}. 







VRMLReader class methods 

   

example1

    "self halt. VRMLReader example1"

    ^self read: 'c:\wilf\junk.wrl'

 

read: aPathName

    "Read the VRML file specified by aPathName."

    ^self new read: aPathName 







VRMLReader methods 



compilerError: aString at: index in: aClass for: anObject

    "Do nothing"



evaluate: aString

    "For VisualSmalltalk"

     ^Compiler evaluate: aString in: UndefinedObject

        to: nil notifying: self ifFail: [nil]

    "For IBM Smalltalk

    ^EsCompiler evaluate: aString for: nil

        ifFail: [:unused | nil]"

    "For VisualWorks

    ^Compiler evaluate: aString"



filteredString: inputStream

    "Return a string with the following replacements:

        {...} by (braces ...)

        [...] by (squares ...)

        ''...'' by '...'

        space .digit by space 0.digit

        0x... by 16r...

        , discarded

        comments discarded"



    | outputStream character |

    inputStream reset.

    outputStream := ReadWriteStream on: (String new: inputStream size).

    outputStream nextPutAll: '#('.

    [inputStream atEnd ] whileFalse:[

        character := inputStream next.

        character == $, ifTrue: [outputStream space "discard"] ifFalse: [

        character == ${ ifTrue: [outputStream nextPutAll: '(braces '] ifFalse: [

        character == $} ifTrue: [outputStream nextPutAll: ')'] ifFalse: [

        character == $[ ifTrue: [outputStream nextPutAll: '(squares '] ifFalse: [

        character == $] ifTrue: [outputStream nextPutAll: ')'] ifFalse: [

        character == $" ifTrue: [outputStream nextPut: $'; nextPutAll: (inputStream upTo: $"); nextPut: $'] ifFalse: [

        character == $# ifTrue: [inputStream nextLine] ifFalse: [

        character == $0 ifTrue: [

            (self isStreamX: inputStream)

                ifTrue: [inputStream next. outputStream nextPutAll: '16r']

                ifFalse: [outputStream nextPut: $0]] ifFalse: [

        character isWhitespace ifTrue: [

            outputStream space.

            (self isStreamDotNumber: inputStream)

                ifTrue: [outputStream nextPut: $0]] ifFalse: [

        outputStream nextPut: character]]]]]]]]]].

    outputStream nextPutAll: ')'.

    ^outputStream contents trimBlanks



initialize

    variables := IdentityDictionary new



isBraceList: item

    (item isKindOf: Array) ifFalse: [^false].

    item isEmpty ifTrue: [^false].

    ^item first == #braces

   

isFieldName: name

    name isSymbol ifFalse: [^false].

    name isEmpty ifTrue: [^false].

    ^name first isLowerCase

   

isName: name

    name isSymbol ifFalse: [^false].

    name isEmpty ifTrue: [^false].

    ^name first isLetter

   

isNodeName: name

    name isSymbol ifFalse: [^false].

    name isEmpty ifTrue: [^false].

    ^name first isUpperCase & (

        name last isLowerCase | name last isDigit)



isNumberOrStringLiteral: item

    item isNumber ifTrue: [^true].

    item isSymbol ifTrue: [^false].

    ^item isString

 

isSquareList: item

    (item isKindOf: Array) ifFalse: [^false].

    item isEmpty ifTrue: [^false].

    ^item first == #squares

 

isStreamDotNumber: stream

    | position isDigit |

    stream atEnd ifTrue: [^false].

    stream peek == $. ifFalse: [^false].

    position := stream position.

    stream next.

    stream atEnd ifTrue: [stream position: position. ^false].

    isDigit := stream peek isDigit.

    stream position: position.

    ^isDigit



isStreamX: stream

    stream atEnd ifTrue: [^false].

    ^stream peek asLowerCase == $x



isVersion1

    ^isVersion1



isVersion1: stream

    ^self streamHeader: stream matches: '#VRML V1.0 ascii'

 

isVersion2: stream

    ^self streamHeader: stream matches: '#VRML V2.0 utf8'

  

nextItem: stream

    ^self perform: (self whatsNext: stream peek) with: stream.

   

printOn: aStream indentNext: indent

    aStream nextPutAll: 'aVRMLReader'

 

processCollection: stream

    | collection item |

    collection := OrderedCollection new.

    [stream atEnd]  whileFalse: [

        (self isFieldName: stream peek) ifTrue: [

            self error: 'Field names ', stream next printString, ' not expected in a collection'].

        item := self nextItem: stream.

        item isCollection

            ifTrue: [collection addAll: item]

            ifFalse: [collection add: item]].

    ^collection asArray

   

processDefinition: stream

    | name value |

    stream next.

    name := stream next.

    (variables includesKey: name) ifTrue: [

        ^self error: 'DEF occurred for second time on ', name printString].

    variables at: name put: (value := self nextItem: stream).

    ^VRMLVariable new

        name: name;

        value: value;

        isDefinition: true

  

processExternProto: nodeStream

    self error: 'external prototype nodes not yet handled'

 

processField: stream

    ^Association

        key: stream next

        value: (self nextItem: stream)

   

processIs: nodeStream

    self error: 'prototype IS not yet handled'

  

processNode: nodeStream

    | node name array pair stream |

    "Prepare for new node."

    name := nodeStream next. array := nodeStream next.

    (self isBraceList: array) ifFalse: [

        self error: 'expected braces after node ', node name printString].



    "Determine if we have a collection (which only VRML1 supports)

    or named fields (which VRML1 and 2 support)."

    stream := ReadStream on: array.

    stream next. "get rid of brace"

    node := VRMLNode new name: name.

    (stream atEnd not and: [self isFieldName: stream peek])

        ifFalse: [

            node at: #children put: (self processCollection: stream).

            ^node].



    "Process field names."

    [stream atEnd]  whileFalse: [

        pair := self nextItem: stream.

        pair isAssociation ifFalse: [

            self error: 'expected a field name in node ', name].

        node at: pair key put: pair value].

    ^node



processProto: nodeStream

    self error: 'prototype nodes not yet handled'



processRoute: nodeStream

    self error: 'event routing not yet handled'

  

processUse: stream

    | name |

    stream next.

    name := stream next.

    ^VRMLVariable new

        name: name;

        value: (variables at: name ifAbsent: [

            self error: 'USE name ', name printString, ' unknown']);

        isDefinition: false



processValue: stream

    | data |

    data := stream next.

    (self isName: data) ifTrue: [^data].

    (self isNumberOrStringLiteral: data) ifTrue: [

        data := OrderedCollection with: data.

        [stream atEnd not and: [self isNumberOrStringLiteral: stream peek]] whileTrue: [

            data add: stream next].

        data size = 1 ifTrue: [^data first] ifFalse: [^data asArray]].

    (self isBraceList: data) ifTrue: [

        self error: 'did not expect braces in this context'].

    (self isSquareList: data) ifTrue: [

        data := data copyFrom: 2 to: data size.

        ^self processCollection: (ReadStream on: data)].

    self error: 'Unexpected data ', data printString

  

read: aPathName

    "Read the VRML file specified by aPathName."

    | stream array |

    stream := File pathNameReadOnly: aPathName.

    (isVersion1 := (self isVersion1: stream)) |

    (self isVersion2: stream) ifFalse: [

        stream close.

        self error: 'This is not a VRML file'.

        ^nil].



    array := self evaluate: (self filteredString: stream).

    stream close.



    array isNil ifTrue: [

        self error: 'Error in VRML file (see transcript)'.

        ^nil].



    stream := ReadStream on: array.

    ^self processCollection: stream

   

streamHeader: stream matches: aString

    "Does the first line of the VRML file match the string?"

    stream size < aString size ifTrue: [^false].

    ^(stream copyFrom: 1 to: aString size) = aString



whatsNext: name

    name == #DEF ifTrue: [^#processDefinition:].

    name == #USE ifTrue: [^#processUse:].

    name == #PROTO ifTrue: [^#processProto:].

    name == #EXTERNPROTO ifTrue: [^#processExternProto:].

    name == #IS ifTrue: [^#processIs:].

    name == #ROUTE ifTrue: [^#processRoute:].

    (self isFieldName: name) ifTrue: [^#processField:].

    (self isNodeName: name) ifTrue: [^#processNode:].

    ^#processValue: 







VRMLVariable class methods  







VRMLVariable methods 



isDefinition

    ^isDefinition



isDefinition: aBoolean

    isDefinition := aBoolean

   

name

    ^name



name: aSymbol

    name := aSymbol

 

printOn: aStream indentNext: indent

    self isDefinition

        ifTrue: [

            aStream

                nextPutAll: 'DEF ';

                nextPutAll: self name;

                nextPutAll: ' = '.

            self privatePrint: self value on: aStream indentNext: indent + 3]

        ifFalse: [

            aStream

                nextPutAll: 'USE ';

                nextPutAll: self name]

   

value

    ^value

  

value: aVRMLObject

    value := aVRMLObject 







Collection methods 



exists: aBooleanBlock

        "Answers true if aBooleanBlock evaluate to

        true for some element; false, otherwise."

    self detect: aBooleanBlock ifNone: [^false].

    ^true 







Object class methods 

 

fileOutOn: fileName classes: classes methods: methods pools: pools

    | stream pool candidate methodName methodType |

    "Get ready."

    "Note: class contains instance methods,

    class class contains class methods."

    stream := File newFile: fileName.



    "Pool dictionaries (they better be simple)."

    pools do: [:poolName |

        pool := Smalltalk at: poolName.

        stream

            nextPutAll: 'Smalltalk at: #'; nextPutAll: poolName;

            nextPutAll: ' put: ('.

        pool storeOn: stream.

        stream nextChunkPut: ')'; cr; cr].



    "Class definitions in an appropriate order."

    classes do: [:className |

        candidate := Smalltalk at: className.

        candidate withAllSubclasses do: [:class |

            class fileOutOn: stream.

            stream nextChunkPut: ''; cr]].



    "Class methods in the class same order."

    classes do: [:className |

        candidate := Smalltalk at: className.

        candidate withAllSubclasses do: [:class |

            (ClassReader forClass: class class) fileOutOn: stream.

            (ClassReader forClass: class) fileOutOn: stream]].



    "Individual methods in individual classes."

    methods do: [:triple |

        methodType := triple first.

        methodName := triple at: 2.

        className := triple last.

        class := Smalltalk at: className.

        methodType == #class "versus #instance"

            ifTrue: [class := class class].

        (ClassReader forClass: class) fileOutMethod: methodName on: stream].



    "Invoke class initialization methods if they exist."

    classes do: [:className |

        candidate := Smalltalk at: className.

        candidate withAllSubclasses do: [:class |

            (class class includesSelector: #initialize) ifTrue: [

                stream cr; cr.

                class printOn: stream.

                stream nextPutAll: ' initialize'.

                stream nextChunkPut: '']]].



    "All done."

    stream close 


